
Figure 1: Alignment of direct effects in MIMIC models

MONTECARLO:
    NAMES = y1-y10 x1-x3;
    NOBSERVATIONS = 1000;
    NREPS = 100;
    categorical=y1-y10;
    generate=y1-y10(3);

ANALYSIS: param=theta; conv=0.000001;

MODEL POPULATION:

    [y1$1-y10$1*-1];
    [y1$2-y10$2*0];
    [y1$3-y10$3*1];
    [x1@0 x2@1 x3@2]; x1@3 x2@2 x3@1;
    f1 by y1-y5*1;
    f2 by y6-y10*1;
    f1 on x1*.5 x2*.6 x3*.7;
    f2 on x1*.7 x2*.6 x3*.5;
    y1 on x3*0.3;
    y2 on x1*0.4;
    f1-f2*.7;
    f1 with f2*.25;
    y1-y10*1;

analysis: tolerance=0.001;

 MODEL:
    [y1$1-y10$1*-1];
    [y1$2-y10$2*0];
    [y1$3-y10$3*1];
    f1 BY y1-y5;              
    f2 BY y6-y10;
    f1 on x1*.5 x2*.6 x3*.7;
    f2 on x1*.7 x2*.6 x3*.5;
    y1 on x3*0.3;
    y2 on x1*0.4;
    y1-y10 ON x1-x3 (d1-d30);
    f1-f2*.7;

     f1 with f2*.25;

MODEL PRIORS: d1-d30~ALF(0,10);

Figure 2: Alignment of cross-loadings simulation study: PSEM-ALF

MONTECARLO:
        NAMES = y1-y6;
 NOBSERVATIONS = 500;
 NREPS = 100;

MODEL POPULATION:
 f1 by y1-y3*1 y6*0.3;
 f2 by y4-y6*1;
 f1-f2@1;
  f1 with f2*.25;
 y1-y6*.5;

  MODEL:
        f1 by y1-y3*1 
          y4-y5*0 y6*0.3 (a1-a3);
 f2 by y4-y6*1
          y1-y3*0 (a4-a6);
 f1-f2@1;
  f1 with f2*.25;
 y1-y6*.5;

MODEL PRIORS: a1-a6~ALF(0,1);

Figure 3: Alignment of cross-loadings simulation study: ESEM-target

MONTECARLO:
        NAMES = y1-y6;
 NOBSERVATIONS = 500;
 NREPS = 100;

MODEL POPULATION:
 f1 by y1-y3*1 y6*0.3;
 f2 by y4-y6*1;
 f1-f2@1;
  f1 with f2*.25;
 y1-y6*.5;

MODEL COVERAGE:
 f1 by y1-y3*1 y6*0.3;
 f2 by y4-y6*1;
 f1-f2@1;
  f1 with f2*.25;
 y1-y6*.5;

analysis: rotation=target;

  MODEL:
       f1 by y1-y3 y4-y6~0 (*1);
       f2 by y1-y3~0 y4-y6 (*1);

Figure 4: Alignment of residual correlations in EFA

   MONTECARLO:
 NAMES = u1-u10;
 NOBSERVATIONS = 500;
 NREPS = 100;

MODEL POPULATION:
    f1 by u1-u5*1; 
    f2 by u3*0.5 u6-u10*1;
    f1 with f2*0.25;
    u1-u10*1 f1-f2*1;
    u1 with u2*0.5;
    u4 with u8*0.3;
    u9 with u10*0.4;

MODEL:
    f1 by u1-u5*1 u6-u10*0 (*1); 
    f2 by u1-u2*0 u3*0.5 u4-u5*0 u6-u10*1 (*1);
    f1 with f2*0.25;
    u1-u10*1;
    u1 with u2*0.5;
    u4 with u8*0.3;
    u9 with u10*0.4;
    u1-u10 with u1-u10 (c1-c45);

MODEL PRIORS: c1-c45~ALF(0,1); 

Figure 5: Alignment of residual correlations and cross-loadings in CFA

   MONTECARLO:
 NAMES = u1-u10;
 NOBSERVATIONS = 500;
 NREPS = 100;

MODEL POPULATION:
    f1 by u1-u5*1; 
    f2 by u3*0.5 u6-u10*1;;
    u1-u10*1 f1-f2*1;
    f1 with f2*0.25;
    u1 with u2*0.5;
    u4 with u8*0.3;
    u9 with u10*0.4;

MODEL:
    f1 by u1-u5*1 
          u6-u10*0 (L1-L5); 
    f2 by u6-u10*1 
          u1-u2*0 u3*0.5 u4-u5*0 (L6-L10);
    u1-u10*1; f1-f2@1;
    f1 with f2*0.25;
    u1 with u2*0.5;
    u4 with u8*0.3;
    u9 with u10*0.4;
    u1-u10 with u1-u10 (c1-c45);

MODEL PRIORS: c1-c45~ALF(0,1); L1-L10~ALF(0,1);

Figure 6: Multiple Group Alignment

MONTECARLO:
          NAMES = y1-y4;
          NOBSERVATIONS = 3(1000);
          NGROUPS = 3; 

     NREPS = 100;        
    
ANALYSIS: alignment=fixed; tolerance=0.001;

MODEL POPULATION: f1 BY y1-y4*1; [y1-y4*0]; y1-y4*.5;

MODEL POPULATION-G1: [y4*0.5]; [f1@0]; f1@1;

MODEL POPULATION-G2: f1 BY y4*0.5; [f1*0.4]; f1*1.5;

MODEL POPULATION-G3: [f1*-0.3]; f1*1.2;

MODEL COVERAGE: f1 BY y1-y4*1; y1-y4*.5;

MODEL COVERAGE-G1: f1*1; [y4*0.5];

MODEL COVERAGE-G2: f1 BY y4*0.5; [f1*0.4]; f1*1.5;

MODEL COVERAGE-G3: [f1*-0.3]; f1*1.2;

MODEL: f1 BY y1-y4;

7



Figure 7: PSEM Multiple Group Alignment

MONTECARLO:
      NAMES = y1-y4;
      NOBSERVATIONS = 3(1000);
      NGROUPS = 3; 
      NREPS = 100;        

MODEL POPULATION: f1 BY y1-y4*1; [y1-y4*0]; y1-y4*.5;

MODEL POPULATION-G1: [y4*0.5]; [f1@0]; f1@1;

MODEL POPULATION-G2: f1 BY y4*0.5; [f1*0.4]; f1*1.5;

MODEL POPULATION-G3: [f1*-0.3]; f1*1.2;

MODEL COVERAGE: f1 BY y1-y4*1; y1-y4*.5;

MODEL COVERAGE-G1: f1*1; [y4*0.5];

MODEL COVERAGE-G2: f1 BY y4*0.5; [f1*0.4]; f1*1.5;

MODEL COVERAGE-G3: [f1*-0.3]; f1*1.2;

MODEL: f1 BY y1-y4*1 (L#_1-L#_4);
      [y1-y4*0] (M#_1-M#_4);       

MODEL G1: [f1@0]; f1@1;

MODEL PRIOR:
DO(1,4) DIFF(L1_#-L3_#)~ALF(0,1);
DO(1,4) DIFF(M1_#-M3_#)~ALF(0,1);

Figure 8: Metric scale alignment

MONTECARLO:
      NAMES = y1-y4;
      NOBSERVATIONS = 3(1000);
      NGROUPS = 3; 

     NREPS = 100;        

MODEL POPULATION: f1 BY y1-y4*1; y1-y4*.5; 

MODEL POPULATION-G1: [f1@0]; f1@1; [y1*-1 y2*1 y3*-0.5 y4*0.5]; 

MODEL POPULATION-G2: f1 BY y4*0.5; f1*1.5; [y1*1 y2*-1 y3*0.5 y4*-0.5]; 

MODEL POPULATION-G3: f1*1.2; [y1-y4*1];

MODEL COVERAGE: f1 BY y1-y4*1; y1-y4*.5;

MODEL COVERAGE-G1: f1@1; [y1*-1 y2*1 y3*-0.5 y4*0.5]; 

MODEL COVERAGE-G2: f1 BY y4*0.5; f1*1.5; [y1*1 y2*-1 y3*0.5 y4*-0.5]; 

MODEL COVERAGE-G3: f1*1.2; [y1-y4*1];

MODEL: f1 BY y1-y4*1 (L#_1-L#_4);
      [y1-y4*0] (M#_1-M#_4); [f1@0];      

MODEL G1: f1@1;

MODEL PRIOR:
DO(1,4) DIFF(L1_#-L3_#)~ALF(0,1);

9



Figure 9: Longitudinal alignment

  MONTECARLO:
      NAMES = y11-y13 y21-y23 y31-y33 y41-y43 Y51-Y53;
      NOBSERVATIONS = 1000;
      NREPS = 100;
 
  analysis: alignment=fixed; tolerance=0.001;

  MODEL POPULATION:
      f1 BY y11-y13*1 ; 
      f2 BY y21*1 y22*0.5 y23*1;
      f3 BY y31-y33*1;
      f4 BY y41-y43*1;
      f5 BY y51-y53*1;
      [y42*0.5];
      f1-f5 with f1-f5*0.5;
      f1*1 f2*0.9 f3*1.4 f4*1.5 f5*2;
      [f1*0 f2*0.3 f3*0.6 f4*0.9 f5*1.2];
      y11-y53*1;

  MODEL:
      f1 BY y11-y13*1 ; 
      f2 BY y21*1 y22*0.5 y23*1;
      f3 BY y31-y33*1;
      f4 BY y41-y43*1;
      f5 BY y51-y53*1;
      [y42*0.5];
      f1-f5 with f1-f5*0.5;
      f1*1 f2*0.9 f3*1.4 f4*1.5 f5*2;
      [f1*0 f2*0.3 f3*0.6 f4*0.9 f5*1.2];
      y11-y53*1;

  MODEL t1:  f1 by y11-y13;
  MODEL t2:  f2 by y21-y23;  
  MODEL t3:  f3 by y31-y33;
  MODEL t4:  f4 by y41-y43;
  MODEL t5:  f5 by y51-y53;

Figure 10: Longitudinal alignment growth model

  MONTECARLO:
      NAMES = y11-y13 y21-y23 y31-y33 y41-y43 Y51-Y53;
      NOBSERVATIONS = 1000;
      NREPS = 100;

  MODEL POPULATION:
      f1 BY y11-y13*1; 
      f2 BY y21*1 y22*0.5 y23*1;
      f3 BY y31-y33*1;
      f4 BY y41-y43*1;
      f5 BY y51-y53*1;
      [y42*0.5];
      f1-f5*1;
      [f1-f5*0];
      y11-y53*1;
      i s | f1@0 f2@1 f3@2 f4@3 f5@4;
      [i*0 s*0.3];
      i*1 s*0.3;
      i with s*-0.2;

  MODEL:
      f1 BY y11-y13*1 (a11-a13); 
      f2 BY y21*1 y22*0.5 y23*1 (a21-a23);
      f3 BY y31-y33*1 (a31-a33);
      f4 BY y41-y43*1 (a41-a43);
      f5 BY y51-y53*1 (a51-a53);
      [y42*0.5];
      [y11-y53] (n1-n15);
      f1-f5*1;
      [f1-f5@0];
      y11-y53*1;
      i s | f1@0 f2@1 f3@2 f4@3 f5@4;
      [i@0 s*0.3];
      i@1 s*0.3;
      i with s*-0.2;

  MODEL PRIOR:
      DIFF(a11 a21 a31 a41 a51)~ALF(0,1);
      DIFF(a12 a22 a32 a42 a52)~ALF(0,1);
      DIFF(a13 a23 a33 a43 a53)~ALF(0,1);
      DIFF(n1 n4 n7 n10 n13)~ALF(0,1);
      DIFF(n2 n5 n8 n11 n14)~ALF(0,1);
      DIFF(n3 n6 n9 n12 n15)~ALF(0,1);

Figure 11: EFA with PSEM

MONTECARLO:
    NAMES = y1-y20;
    NOBSERVATIONS = 1000;
    NREPS = 100;

MODEL POPULATION:
    f1 BY y1*0.7 y2*0.3 y3-y4*0.4 y5*0.3;
    f2 BY y5*0.6 y6*0.7 y7*0.5 y8*0.5 y9*1 y10*0.5;
    f3 BY y11*0.7 y12*0.3 y13-y15*0.4;
    f4 BY y15*0.6 y16*0.7 y17*0.5 y18*0.5 y19*1 y20*0.5;
    f1-f4@1;
    y1-y4*0.35 y5-y8*0.42 y9-y10*0.51;
    y11-y14*0.35 y15-y18*0.42 y19-y20*0.51;
    f1 with f2*0.4; f1 with f3*0.1;    
    f1 with f4*0.3; f2 with f3*0.3;
    f2 with f4*0.2; f3 with f4*0.2;

MODEL:
    f1 BY y1*0.7 y2*0.3 y3-y4*0.4 y5*0.3 y6-y20*0 (a1-a20);
    f2 BY y1-y4*0 y5*0.6 y6*0.7 y7*0.5 y8*0.5 y9*1 y10*0.5 y11-y20*0 (a21-a40);
    f3 BY y1-y10*0 y11*0.7 y12*0.3 y13-y14*0.4 y15*0.4 y16-y20*0 (a41-a60);
    f4 BY y1-y14*0 y15*0.6 y16*0.7 y17*0.5 y18*0.5 y19*1 y20*0.5 (a61-a80);
    f1-f4@1;
    y1-y4*0.35 y5-y8*0.42 y9-y10*0.51;
    y11-y14*0.35 y15-y18*0.42 y19-y20*0.51;
    f1 with f2*0.4; f1 with f3*0.1;    
    f1 with f4*0.3; f2 with f3*0.3;
    f2 with f4*0.2; f3 with f4*0.2;

MODEL PRIORS: 
a1-a80~Geomin(4,.1);

Figure 12: PSEM-ESEM single data input file

variable: NAMES = y1-y20;
   
data: FILE=PSEMESEM1.dat;

analysis: iter=10000;

MODEL:
    f1 BY y1-y20*1 (a1-a20);
    f2 BY y1-y20*1 (a21-a40);
    f3 BY y1-y20*1 (a41-a60);
    f4 BY y1-y20*1 (a61-a80);
    f1-f4@1;

MODEL PRIORS: a1-a80~Geomin(4,.1);

Figure 13: Hierarchical EFA

MONTECARLO:
    NAMES = y1-y20;
    NOBSERVATIONS = 2000;
    NREPS = 100;

MODEL POPULATION:
    f1 BY y1*0.7 y2*1.3 y3-y4*0.8 y5*0.3;
    f2 BY y5*0.6 y6*0.7 y7*0.5 y8*0.8 y9*1 y10*0.5 ;
    f3 BY y11*0.7 y12*1 y13-y15*0.4;
    f4 BY y15*0.6 y16*0.7 y17*1 y18*0.5 y19*1 y20*0.5;
    f1-f4@1;
    y1-y20*1;
    f0 by f1*0.4 f2*0.5 f3*0.5 f4*0.6; f0@1;

MODEL:
    f1 BY y1*0.7 y2*1.3 y3-y4*0.8 y5*0.3 y6-y20*0(a1-a20);
    f2 BY y1-y4*0 y5*0.6 y6*0.7 y7*0.5 y8*0.8 y9*1 y10*0.5 y11-y20*0 (a21-a40);
    f3 BY y1-y10*0 y11*0.7 y12*1 y13-y14*0.4 y15*0.4 y16-y20*0 (a41-a60);
    f4 BY y1-y14*0 y15*0.6 y16*0.7 y17*1 y18*0.5 y19*1 y20*0.5 (a61-a80);
    f1-f4@1;
    y1-y20*1;
    f0 by f1*0.4 f2*0.5 f3*0.5 f4*0.6; f0@1;

MODEL PRIORS: a1-a80~Geomin(4,0.1,0.001);

Figure 14: Generalized bi-factor EFA

MONTECARLO:
NAMES = y1-y20;
NOBSERVATIONS = 1000;
NREPS = 100;

MODEL POPULATION:    
g1 BY y1*.7 y2*.3 y3-y4*.4 y5-y10*1 y11*.2 y12*.3;
g2 BY y11-y20*1;
f1 BY y5*.6 y6*.7 y7*.5 y18*.4 y19*.3 y20*.4;
f2 BY y11*.7 y12*.3 y13-y14*.4 y1-y4*1 y20*.2;
f3 BY y15*.6 y16*.7 y17*.5 y8*.5 y9*1 y10*.5;
y1-y4*.35 y5-y8*.42 y9-y10*.51;
y11-y14*.35 y15-y18*.42 y19-y20*.51;
f1-f3@1; f1-f3 with f1-f3@0;
g1-g2@1; g1-g2 with f1-f3@0;
g1 with g2*.4;    

MODEL:
g1 BY y1*.7 y2*.3 y3-y4*.4 y5-y10*1 y11*.2 y12*.3 y13-y20*0 (a1-a20);
g2 BY y1-y10*0 y11-y20*1 (a21-a40);
f1 BY y1-y4*0 y5*.6 y6*.7 y7*.5 y8-y17*0 y18*.4 y19*.3 y20*.4 (a41-a60);
f2 BY y1-y4*1 y5-y10*0 y11*.7 y12*.3 y13-y14*.4 y15-y19*0 y20*.2 (a61-a80);
f3 BY y1-y7*0 y8*.5 y9*1 y10*.5 y11-y14*0 y15*.6 y16*.7 y17*.5 y18-y20*0 (a81-a100);
y1-y4*.35 y5-y8*.42 y9-y10*.51;
y11-y14*.35 y15-y18*.42 y19-y20*.51;
f1-f3@1; f1-f3 with f1-f3@0;
g1-g2@1; g1-g2 with f1-f3@0;
g1 with g2*.4;
  
MODEL PRIORS: a1-a100~Geomin(5,1,.001); 

Figure 15: Scalar invariant orthogonal EFA

montecarlo:
 names = y1-y6;
 ngroups = 2;
 nobs = 200 100;
 nreps = 100;

model population:
 f1 by y1-y3*1; y1-y3@1;
 f2 by y4-y6*1 y1*0.3; y4-y6@1;
 f1-f2@1;
 
model population-g2:
 f1*1.3; [f1*0.4];
 f2*1.4; [f2*0.9];

model:
 f1 by y1-y3*1 Y4-Y6*0 (a1-a6); 
 f2 by y1*0.3 Y2-Y3*0 y4-y6*1 (a7-a12); 
        f1 with f2@0; f1-f2@1; y1-y6*1;

model g2:
 f1*1.3; [f1*0.4];
 f2*1.4; [f2*0.9];
 
model prior: a1-a12~Geomin(2,1,.001) 

Figure 16: Partial invariance EFA

montecarlo:
    names = y1-y6;
    ngroups = 2;
    nobs = 300 300;
    nreps = 100;

model population:
    f1 by y1-y3*1; y1-y3@1;
    f2 by y4-y6*1 y1*0.3; y4-y6@1;
    f1-f2@1; f1 with f2*0.3;
 
model population-g2:
    f1*1.3 f2*1.4; f1 with f2*0.1; 
    [f1*0.4 f2*0.9];
    f2 by Y4*0.5; ! non-invariant loading
    [y6*0.5]; ! non-invariant intercept
        
model:
    f1 by y1-y3*1 Y4-Y6*0 (a1-a6); 
    f2 by y1*0.3 Y2-Y3*0 y4-y6*1 (a7-a12); 
    f1 with f2*0.3; f1-f2@1; y1-y6*1;

model g2:
    f1*1.3 f2*1.4; f1 with f2*0.1; 
    [f1*0.4 f2*0.9];
    f2 by Y4*0.5; ! non-invariant loading
    [y6*0.5]; ! non-invariant intercept
 
model prior: a1-a12~Geomin(2,1,.001);

17



Figure 17: ELGM Mplus input files

 TITLE:      ELGM with 4 factors and T=12
DATA:        FILE IS 1.dat;
VARIABLE:    NAMES ARE y1-y12;

 MODEL:      f1-f4 BY y1-y12 (*1);
             [f1-f4 y1-y12@0];

----------------------------------

 TITLE:      ELGM-null with 4 factors and T=12
 DATA:      FILE IS 1.dat;

VARIABLE:    NAMES ARE y1-y12;
 MODEL:      f1 BY y1-y12*1;

             f2 BY y2-y12*1;
             f3 BY y3-y12*1;
             f4 BY y4-y12*1;
             [f1-f4 y1-y12@0];
             f1-f4 with f1-f4@0;
             f1-f4@1;

----------------------------------

 TITLE:      ELGM null with auto-correlation, 4 factors and T=12  
 DATA:      FILE IS 1.dat;

VARIABLE:    NAMES ARE y1-y12;
ANALYSIS:    STARTS=30; iteration=10000;

 MODEL:      f1 BY y1-y12*1;
             f2 BY y2-y12*1;
             f3 BY y3-y12*1;
             f4 BY y4-y12*1;
             [f1-f4 y1-y12@0];
             f1-f4 with f1-f4@0;
             f1-f4@1;
             y2^-y12^ pon y1^-y11^ (r);

Figure 18: PSEM-ELGM linear growth simulation study

MONTECARLO: 
NAMES = y1-y8;
NOBSERVATIONS = 1000;
NREPS = 100;

ANALYSIS: conv=0.000001; iter=100000;

MODEL POPULATION:
y1-y8*1;
I BY Y1*1 Y2*1.03 Y3*1.06 Y4*0.89 Y5*0.94 Y6*0.97 Y7*1 Y8*1.06 (a1-a8);
S BY Y1*0 Y2*0.1 Y3*0.19 Y4*0.29 Y5*0.48 Y6*0.58 Y7*0.68 Y8*0.86 (b1-b8);
I WITH S*0.3; [I*1. S*0.5]; I@1 S@1; 
[y2*0.2 y6*0.4];
y2^-y8^ pon y1^-y7^*0.3(r);

MODEL:
y1-y8*1;
I BY Y1*1 Y2*1.03 Y3*1.06 Y4*0.89 Y5*0.94 Y6*0.97 Y7*1 Y8*1.06 (b1-b8);
S BY Y1*0 Y2*0.1 Y3*0.19 Y4*0.29 Y5*0.48 Y6*0.58 Y7*0.68 Y8*0.86 (g1-g8);
I WITH S*0.3; [I*1. S*0.5]; I@1 S@1; 
[y1*0 y2*0.2 y3-y5*0 y6*0.4 y7-y8*0] (a1-a8); 
y2^-y8^ pon y1^-y7^*0.3(r);

MODEL PRIOR:
a1-a8~ALF(0,1);
DIFF(b1-b8)~ALF(0,1);
g1~ALF(0,1);

Figure 19: Height empirical example

DATA: file=1.dat;

VARIABLE: names=y1-y14; 

ANALYSIS: iter=100000; starts=30;

MODEL:
[y1-y14] (a1-a14);
f1 by y1-y14*1 (b1-b14);
f2 by y1-y14*1 (g1-g14);
y2^-y14^ pon y1^-y13^(r);
f1-f2@1; [f1 f2];

MODEL CONSTRAINT: 
new(d1-d13);
d1=b2-b1;
d2=b3-b2;
d3=b4-b3;
d4=b5-b4;
d5=b6-b5;
d6=b7-b6;
d7=b8-b7;
d8=b9-b8;
d9=b10-b9;
d10=b11-b10;
d11=b12-b11;
d12=b13-b12;
d13=b14-b13;

MODEL PRIOR:
a1-a14~ALF(0,1);
DIFF(d1-d13)~ALF(0,1);
g1~ALF(0,1);

Figure 20: PSEM-ELGM simulation study: mimicking the height data

MONTECARLO: 
NAMES = y1-y10;
NOBSERVATIONS = 1000;
NREPS = 100;

ANALYSIS: conv=0.00001; iter=100000;

MODEL POPULATION:
[y1-y9*0 y10*-0.2]; 
F1 BY Y1*1 Y2*1.1 Y3*1.2 Y4*1.3 Y5*1.4 Y6*1.5 Y7*1.6 Y8*1.7 Y9*1.8 y10*1.9;
F2 BY Y1-Y5*0 Y6*0.4 Y7*0.8 Y8*1.2 y9*0.8 y10*0.4;
F1 WITH F2*0.3; [F1*1. F2*0.5]; F1@1 F2@1; 
y2^-y10^ pon y1^-y9^*0.5(r);
y1-y10*1;

MODEL:
[y1-y9*0 y10*-0.2] (a1-a10); 
F1 BY Y1*1 Y2*1.1 Y3*1.2 Y4*1.3 Y5*1.4 Y6*1.5 Y7*1.6 Y8*1.7 Y9*1.8 y10*1.9 (b1-b10);
F2 BY Y1-Y5*0 Y6*0.4 Y7*0.8 Y8*1.2 y9*0.8 y10*0.4 (g1-g10);
F1 WITH F2*0.3; [F1*1. F2*0.5]; F1@1 F2@1; 
y2^-y10^ pon y1^-y9^*0.5(r);
y1-y10*1;

MODEL CONSTRAINT: 
new(d1-d9);
d1=b2-b1;
d2=b3-b2;
d3=b4-b3;
d4=b5-b4;
d5=b6-b5;
d6=b7-b6;
d7=b8-b7;
d8=b9-b8;
d9=b10-b9;

MODEL PRIOR:
a1-a10~ALF(0,1);
DIFF(d1-d9)~ALF(0,1);
g1~ALF(0,1);

